/* Load the dataset into SAS and name it ‘MixMF’.

This SAS code fits the model on which the results are based in the manuscript ‘Effects of experimental drought and plant diversity on ecosystem multifunctionality of a model system for crop rotation’ by Guylain Grange, Caroline Brophy, Rishabh Vishwakarma, John Finn
Fixed effects include species identity effects (species proportions) crossed with drought and ecosystem function, and within- and between-functional group interactions only where evidenced in univariate models.
The variable XN distinguishes the plots that received the high N fertiliser rate (300N, only for Lolium perenne monocultures) and is also crossed with drought and ecosystem function. When N fertiliser rate is high, XN=1 and Lp=0, in all other cases XN=0; thus, the fitted coefficients of XN are the estimated averages of the high N Lolium perenne monoculture plots for each water supply level and function. Because the species proportions sum to 1, the intercept is omitted to avoid over parameterisation. 
	The random statement accounts for the split-plot structure within each function. 
	The repeated statement accounts for repeated measures for the six ecosystem functions within each split plot.
*/
/************************************************************/
/* Diversity-Interactions model including six ecosystem functions */
/************************************************************/
/*Sort data by subject for modelling on repeated measures */
proc sort data=MixMF;
	by split plot func;
run;

/* Fit the model*/
/*A store command is added at the end. This will store the fitted model and can be reused to test the estimate statements without having to fit the model again. The model can take about one hour to run.*/

proc mixed data=MixMF method=reml /*plots=all*/;
      class droughtInt plot split func;
      model Score = XN*func*DroughtInt Lp*func*DroughtInt Pp*func*DroughtInt Tp*func*DroughtInt Tr*func*DroughtInt Ci*func*DroughtInt Pl*func*DroughtInt
                        Lp*Pp*func*SpInt Tp*Tr*func*SpInt Ci*Pl*func*SpInt
                        Grass*Legume*func*SpInt Grass*Herb*func*SpInt Legume*Herb*func*SpInt

                        / noint solution ddfm=kr;
      random plot / group=func;
      repeated func / subject=split(plot) type=un r rcorr;
         store BestModel / label='Best Model';  
run;



proc plm restore=BestModel;
	show parameters;
run;


/* Some examples of the estimate statements used to obtain values*/
proc plm restore=BestModel;

    estimate 'Yield consistency, average of all monocultures, rainfed' Lp*func*DroughtInt 0.16666666667 0 0 0 0 0 0 0 0 
                                            Pp*func*DroughtInt 0.16666666667 0 0 0 0 0 0 0 0 
                                            Tr*func*DroughtInt 0.16666666667 0 0 0 0 0 0 0 0 
                                            Tp*func*DroughtInt 0.16666666667 0 0 0 0 0 0 0 0 
                                            Ci*func*DroughtInt 0.16666666667 0 0 0 0 0 0 0 0 
                                            Pl*func*DroughtInt 0.16666666667 0 0 0 0 0 0 0 0;


